
MCS 320 Introduction to Symbolic Computation Spring 2007

Maple Lecture 26. Solving Equations

This lecture is inspired on [1, Chapter 16]. We first consider equations in one unknown, emphasizing the
role of parameters. We apply the command solve systems of linear and polynomial equations. This lecture
ends with mentioning some special solvers.

26.1 Equations in one unknown

The main command to solve equations is solve:

[> equ := a^2*(x^2+x+1)-a*(2*x-3)-x^2-3*x+2;

[> sols := solve(equ,x);

Maple returns us two roots, but we have to be careful: the denominator in the solutions may vanish. We
investigate what the equation looks like for this special parameter value:

[> equ_a1 := subs(a=1,equ);

When the parameter a equals one, then there is only one solution:

[> solve(equ_a1,x);

We are not done yet. There are special values of the parameters for which there are fewer solutions (as
above), but we can also have more than two solutions. Here is how we can find this out:

[> solve(equ,a);

Switching roles of a and x, we see there is a solution for a that does not depend on x:

[> equ_am1 := subs(a=-1,equ);

For a = −1, the equation vanishes entirely and there are infinitely many values for x.

26.2 Some difficulties interpreting the output of solve

1) Output of solve needs further processing Consider

[> eq := x=cos(x);

[> s := solve(eq,x);

We have seen the RootOf with algebraic numbers. We can approximate:

[> evalf(s);

or solve directly numerically:

[> fsolve(eq,x);

2) Too few solutions Sometimes Maple does not recognize the periodicity.

[> eq := sin(x) = 1/2;

[> solve(eq,x);

We can change the environment variable EnvAllSolutions to true:

[> _EnvAllSolutions := true;

[> sols := solve(eq,x);

The B1 denotes a binary number (0 or 1), and Z1 denotes an arbitrary integer. Let us pick a solution:

[> asol := subs({_B1 = 1,_Z1 = 12},sols);

[> sin(asol);

3) Too many solutions Consider the following equation:

[> eqn := (sin(x))^3 - 3*sin(x) + 1 = 3;

[> sols := solve(eqn,x);

The above solution is not so useful: observe that there is repetition in the sequence above. Note the
arcsin(2).... We better recognized the polynomial structure of the equation:

[> solve(eqn,sin(x));

We see that −1 is a double root of the equation:

[> expand((sin(x)-2)*(sin(x)+1)^2) = lhs(eqn) - rhs(eqn);

26.3 Systems of Equations

Maple can solve systems of linear equations symbolically :

[> eq1 := A[1,1]*x[1] + A[1,2]*x[2] = b[1];

[> eq2 := A[2,1]*x[1] + A[2,2]*x[2] = b[2];

[> solve({eq1,eq2},{x[1],x[2]});

We recognize that Maple uses Cramer’s rule to solve this linear system. The denominator in the root is
the determinant of the matrix A. Just like in the case of solving equations with parameters, we ought to be
careful when interpreting and using this formal result: a system of linear equations can have exactly one,
infinitely many or no solutions at all.

26.4 The Gröbner basis method

The Gröbner basis method can be seen as a generalization of the elimination method for linear systems and
greatest common divisor of polynomials in one variable. One of its application is the solution of polynomial
systems.

[> p1 := x^2 + x^3 - y^2; p2 := (x-1)^2 + y^2 - 1;

We wish to find the common intersection points of the cubic (defined by p1) with the circle (defined by p2).
For this two dimensional problem we can visualize the intersection problem:

[> with(plots):

[> plot1 := implicitplot(p1,x=-2..2,y=-2..2,color=red):

[> plot2 := implicitplot(p2,x=-2..2,y=-2..2,color=green):

[> display(plot1,plot2,scaling=constrained);

A Gröbner basis with pure lexicographic order produces an equivalent triangular system:

[> with(grobner);

[> gb := gbasis({p1,p2},[x,y],‘plex‘);

To get the y-coordinates of the solutions, we compute the roots of the second polynomial in gb:

[> ysols := solve(gb[2],y);

We see the origin appearing as a double root (this explains why Maple has troubles displaying the cubic at
the origin). To get the corresponding x-coordinates of the solutions, we substitute the y-coordinate in the
first equation of gb. Observe that gb[1] is linear in x: for every value for y, we get exactly one value for x.

[> x3 := subs(y=ysols[3],gb[1]);

[> solve(x3,x);

So we get six solutions: the origin as double root. Two real solutions we can see as intersection points on the
plot, and two complex conjugated solutions.

26.5 Numerical and other solvers in Maple

To find numerical approximations, we use fsolve. Others solvers are isolve (to find integer roots), msolve
(for modular arithmetic), and rsolve (to solve recurrence relations). For example, the command

[> rsolve({f(n+1) = f(n) + f(n-1),f(0)=0,f(1)=1},f(n));

gives an explicit formula for the n-th Fibonacci number.

26.6 Assignments

1. Consider the polynomial equation x2a2 − 12x2 + a2x− 3
√
3ax + 6x + a2 −

√
3a− 6 = 0.

Give the Maple command to solve the equation with x as variable.

For which values of the parameter a does the equation have less than two solutions?

Find the solution for this special value of the parameter.

For which values of the parameter a does the equation have infinitely many solutions?

Give the Maple commands you used to test your answer.

2. Solve the equation x = 4
x+3

.

Use fsolve, starting at some value, i.e.: use fsolve as fsolve(equation,x=0.7).

For different values at the start we may find different roots.

Determine which values at the start give which roots.

3. Solve the system
{

x2 + y2 − 5 = 0
9x2 + y2 − 9 = 0

Make a plot of the two algebraic curves to confirm the values found for the intersection points.

How many solutions did you find? How many solutions do you see?

4. The cost T (n) of a merge sort on a list of n numbers is governed by the following recurrence relation:

T (n) = 2T
(n

2

)

+ n− 1, T (1) = 0.

Use rsolve to find an explicit solution for T (n). Simplify so you can see T (n) is O(n log(n)).

5. For which x is x3 − x2 + 9 > 0? Give the limits of the intervals as floating-point numbers.

